




Εθνικό Μετσόβιο Πολυτεχνείο
Σχολή Ηλεκτρολόγων Μηχανικών
& Μηχανικών Υπολογιστών
Τομέας Τεχνολογίας Πληροφορικής & Υπολογιστών
Εργαστήριο Τεχνολογίας Λογισμικού


Μεταγλωττιστές 2020

Θέμα εργασίας

Η γλώσσα Tony



Sir Charles Antony (Tony) Richard Hoare (1934–)
Prof. Emeritus, Oxford University



Μεταγλωττιστές

<https://courses.softlab.ntua.gr/compiler/>

Διδάσκοντες: Νίκος Παπασπύρου
Κωστής Σαγώνας

Βοηθός: Βίκυ Κοντουρά

Αθήνα, Μάρτιος 2020

ΘΕΜΑ:

Να σχεδιαστεί και να υλοποιηθεί από κάθε ομάδα, το πολύ δυο σπουδαστών, ένας μεταγλωττιστής για τη γλώσσα Tony. Γλώσσα υλοποίησης μπορεί να είναι μια από τις C/C++, Java, SML, OCaml, Haskell ή Python, αλλά έχετε υπ' όψη σας ότι κομμάτια του μεταγλωττιστή και εργαλεία που μπορεί να σας φανούν χρήσιμα μπορεί να μην είναι διαθέσιμα σε κάποιες από τις παραπάνω γλώσσες και σε αυτήν την περίπτωση θα πρέπει να τα υλοποιήσετε μόνοι σας. Η επιλογή κάποιας άλλης γλώσσας υλοποίησης μπορεί να γίνει κατόπιν συνεννόησης με τους διδάσκοντες. Επιτρέπεται να χρησιμοποιηθούν και επίσης συνιστάται η χρήση εργαλείων, π.χ. flex/ML-Lex/ocamllex/Alex, bison/ML-Yacc/ocamlyacc/Happy, JavaCC, κ.λπ. Περισσότερες πληροφορίες σχετικές με αυτά τα εργαλεία θα δοθούν στις παραδόσεις.

Παραδοτέα, ημερομηνίες και βαθμολόγηση

Τα τμήματα του μεταγλωττιστή και η κατανομή μονάδων φαίνονται στον παρακάτω πίνακα. Οι ημερομηνίες παράδοσης αναγράφονται στη σελίδα του μαθήματος.

<i>Τμήμα του μεταγλωττιστή</i>	<i>Μονάδες</i>	<i>Bonus</i>
Λεκτικός αναλυτής	0.5	—
Συντακτικός αναλυτής	0.5	—
Σημασιολογικός αναλυτής	1.0	—
Ενδιάμεσος κώδικας	1.0	—
Βελτιστοποίηση	1.0	1.0
Τελικός κώδικας	1.0	1.0
<i>Συνολική εργασία και έκθεση</i>	5.0	2.0

Για τα διάφορα τμήματα της εργασίας πρέπει να παραδίδεται εμπρόθεσμα από κάθε ομάδα ο αντίστοιχος κώδικας σε ηλεκτρονική μορφή, καθώς και σαφείς οδηγίες για την παραγωγή ενός εκτελέσιμου προγράμματος επίδειξης της λειτουργίας του αντίστοιχου τμήματος, από τον κώδικα αυτόν. Καθυστερημένες ασκήσεις θα βαθμολογούνται με μικρότερο βαθμό, αντιστρόφως ανάλογα προς το χρόνο καθυστέρησης. Παρακαλούμε, **μην παραδίδετε τυπωμένες εργασίες!** Η μορφή και τα περιεχόμενα των παραδοτέων, συμπεριλαμβανομένης και της τελικής έκθεσης, πρέπει να συμφωνούν με τις οδηγίες που δίνονται στην ενότητα 4 του παρόντος.



Προαιρετικά τμήματα και μονάδες bonus

Το σύνολο των μονάδων της παρούσας εργασίας είναι 7. Από αυτές, οι 2 μονάδες είναι bonus (που σημαίνει ότι το σύνολο μονάδων του μαθήματος είναι 12) και αντιστοιχούν στα παρακάτω τμήματα της εργασίας, που είναι προαιρετικά:

- (25%) Βελτιστοποίηση ενδιάμεσου κώδικα, με ανάλυση ροής δεδομένων και ελέγχου.
- (25%) Δέσμευση καταχωρητών και βελτιστοποίηση τελικού κώδικα.
- (50%) Υλοποίηση ενός συλλέκτη σκουπιδιών στο σύστημα εκτέλεσης του μεταγλωττιστή σας.



Περιεχόμενα

1 Περιγραφή της γλώσσας Tony	5
1.1 Λεκτικές μονάδες	5
1.2 Τύποι δεδομένων	6
1.3 Δομή του προγράμματος	7
1.3.1 Μεταβλητές	7
1.3.2 Δομικές μονάδες	7
1.4 Εκφράσεις	8
1.4.1 L-values	8
1.4.2 Σταθερές	8
1.4.3 Τελεστές	8
1.4.4 Κλήση δομικών μονάδων ως συναρτήσεων	9
1.5 Εντολές	10
1.6 Βιβλιοθήκη έτοιμων συναρτήσεων	11
1.6.1 Είσοδος και έξοδος	11
1.6.2 Συναρτήσεις μετατροπής	11
1.6.3 Συναρτήσεις διαχείρισης συμβολοσειρών	11
2 Πλήρης γραμματική της Tony	11
3 Παραδείγματα	12
3.1 Πες γεια!	12
3.2 Οι πύργοι του Hanoi	12
3.3 Πρώτοι αριθμοί	13
3.4 Αντιστροφή συμβολοσειράς	15
3.5 Ταξινόμηση πίνακα με τη μέθοδο της φυσαλίδας	15
3.6 Ταξινόμηση λίστας με τη μέθοδο της διαμέρισης	16
4 Οδηγίες για την παράδοση	18

1 Περιγραφή της γλώσσας Tony

Η γλώσσα Tony είναι μια απλή γλώσσα προστακτικού προγραμματισμού. Τα κύρια χαρακτηριστικά της εν συντομία είναι τα εξής:

- Απλή δομή και σύνταξη εντολών και εκφράσεων που μοιάζει με αυτήν κάποιων γλωσσών σεναρίων.
- Βασικοί τύποι δεδομένων για λογικές τιμές, χαρακτήρες, ακέραιους αριθμούς, μονοδιάστατους πίνακες και λίστες.
- Απλές συναρτήσεις, πέρασμα κατ' αξία ή κατ' αναφορά.
- Εμβέλεια μεταβλητών όπως στην Pascal.
- Βιβλιοθήκη συναρτήσεων.

Περισσότερες λεπτομέρειες της γλώσσας δίνονται στις παραγράφους που ακολουθούν.

1.1 Λεκτικές μονάδες

Οι λεκτικές μονάδες της γλώσσας Tony χωρίζονται στις παρακάτω κατηγορίες:

- Τις λέξεις κλειδιά, οι οποίες είναι οι παρακάτω:

and	bool	char	decl	def	else	elsif
end	exit	false	for	head	if	int
list	mod	new	nil	nil?	not	or
ref	return	skip	tail	true		

- Τα ονόματα, τα οποία αποτελούνται από ένα γράμμα του λατινικού αλφαβήτου, πιθανώς ακολουθούμενο από μια σειρά γραμμάτων, δεκαδικών ψηφίων, χαρακτήρων υπογράμμισης (underscore) ή (λατινικών) ερωτηματικών. Τα ονόματα δεν πρέπει να συμπίπτουν με τις λέξεις κλειδιά που αναφέρθηκαν παραπάνω. Πεζά και κεφαλαία γράμματα θεωρούνται διαφορετικά.
- Τις ακέραιες σταθερές χωρίς πρόσημο, που αποτελούνται από ένα ή περισσότερα δεκαδικά ψηφία. Παραδείγματα ακέραιων σταθερών είναι τα ακόλουθα:

```
0      42      1284      00200
```

- Τους σταθερούς χαρακτήρες, που αποτελούνται από ένα χαρακτήρα μέσα σε απλά εισαγωγικά. Ο χαρακτήρας αυτός μπορεί να είναι οποιοσδήποτε κοινός χαρακτήρας ή ακολουθία διαφυγής (escape sequence). Κοινοί χαρακτήρες είναι όλοι οι εκτυπώσιμοι χαρακτήρες πλην των απλών και διπλών εισαγωγικών και του χαρακτήρα \ (backslash). Οι ακολουθίες διαφυγής ξεκινούν με το χαρακτήρα \ (backslash) και περιγράφονται στον πίνακα 1. Παραδείγματα σταθερών χαρακτήρων είναι οι ακόλουθες:

```
'a'      '1'      '\n'      '\"'      '\x1d'
```

- Τις σταθερές συμβολοσειρές, που αποτελούνται από μια ακολουθία κοινών χαρακτήρων ή ακολουθιών διαφυγής μέσα σε διπλά εισαγωγικά. Οι συμβολοσειρές δεν μπορούν να εκτείνονται σε περισσότερες από μια γραμμές προγράμματος. Παραδείγματα σταθερών συμβολοσειρών είναι οι ακόλουθες:

```
"abc"      "Route66"      "Helloworld!\n"  
"Name:\t\"DouglasAdams\"\\nValue:\t42\n"
```

- Τους συμβολικούς τελεστές, οι οποίοι είναι οι παρακάτω:

Πίνακας 1: Ακολουθίες διαφυγής (escape sequences).

Ακολουθία διαφυγής	Περιγραφή
<code>\n</code>	ο χαρακτήρας αλλαγής γραμμής (line feed)
<code>\t</code>	ο χαρακτήρας στηλοθέτησης (tab)
<code>\r</code>	ο χαρακτήρας επιστροφής στην αρχή της γραμμής (carriage return)
<code>\0</code>	ο χαρακτήρας με ASCII κωδικό 0
<code>\\</code>	ο χαρακτήρας <code>\</code> (backslash)
<code>\'</code>	ο χαρακτήρας <code>'</code> (απλό εισαγωγικό)
<code>\"</code>	ο χαρακτήρας <code>"</code> (διπλό εισαγωγικό)
<code>\xnn</code>	ο χαρακτήρας με ASCII κωδικό <code>nn</code> στο δεκαεξαδικό σύστημα

1 + - * / # = <> < > <= >=

2 • Τους *διαχωριστές*, οι οποίοι είναι οι παρακάτω:

3 () [] , ; : :=

4 Εκτός από τις λεκτικές μονάδες που προαναφέρθηκαν, ένα πρόγραμμα Tony μπορεί επίσης να περιέχει
5 τα παρακάτω, τα οποία διαχωρίζουν λεκτικές μονάδες και αγνοούνται:

6 • *Κενούς χαρακτήρες*, δηλαδή ακολουθίες αποτελούμενες από κενά διαστήματα (space), χαρακτήρες
7 στηλοθέτησης (tab), χαρακτήρες αλλαγής γραμμής (line feed) ή χαρακτήρες επιστροφής στην αρχή
8 της γραμμής (carriage return).

9 • *Σχόλια μιας γραμμής*, τα οποία αρχίζουν με το χαρακτήρα % και τερματίζονται με το τέλος της
10 τρέχουσας γραμμής.

11 • *Σχόλια πολλών γραμμών*, τα οποία αρχίζουν με την ακολουθία χαρακτήρων `<*` και τερματίζονται
12 με την ακολουθία χαρακτήρων `>*`. Τα σχόλια αυτής της μορφής επιτρέπεται να είναι φωλιασμένα.

13 1.2 Τύποι δεδομένων

14 Η Tony υποστηρίζει τρεις βασικούς τύπους δεδομένων:

- 15 • `int`: ακέραιοι αριθμοί μεγέθους τουλάχιστον 16 bit (−32768 έως 32767),
- 16 • `char`: χαρακτήρες, και
- 17 • `bool`: λογικές τιμές.

18 Εκτός από τους βασικούς τύπους, η Tony υποστηρίζει επίσης δύο σύνθετους τύπους:

- 19 • *πίνακες*, που συμβολίζονται με `t []`, όπου `t` έγκυρος τύπος, και
- 20 • *λίστες*, που συμβολίζονται με `list [t]`, όπου `t` έγκυρος τύπος.

21 Οι πίνακες και οι λίστες στην Tony έχουν σημασιολογία αναφοράς: με ανάθεση και με πέρασμα παρα-
22 μέτρου ποτέ δεν αντιγράφονται τα περιεχόμενα ενός πίνακα ή μίας λίστας, αλλά απλώς αντιγράφεται η
23 διεύθυνση της μνήμης όπου είναι αποθηκευμένος ο πίνακας ή η λίστα. Το σύστημα χρόνου εκτέλεσης
24 αναλαμβάνει την αυτόματη αποδέσμευση των πινάκων και των λιστών, υλοποιώντας κάποιον αλγόριθμο
25 συλλογής σκουπιδιών.

1 1.3 Δομή του προγράμματος

2 Η γλώσσα Tony είναι μια δομημένη (block structured) γλώσσα. Ένα πρόγραμμα έχει χοντρικά την ίδια
3 δομή με ένα πρόγραμμα Pascal. Οι δομικές μονάδες μπορούν να είναι φωλιασμένες η μία μέσα στην
4 άλλη και οι κανόνες εμφάνισης είναι οι ίδιοι με αυτούς της Pascal. Το κύριο πρόγραμμα είναι μία δομική
5 μονάδα που δεν επιστρέφει αποτέλεσμα και δε δέχεται παραμέτρους.

6 Κάθε δομική μονάδα μπορεί να περιέχει προαιρετικά:

- 7 • Δηλώσεις μεταβλητών.
- 8 • Ορισμούς υποπρογραμμάτων.
- 9 • Δηλώσεις υποπρογραμμάτων, οι ορισμοί των οποίων θα ακολουθήσουν.

10 1.3.1 Μεταβλητές

11 Οι δηλώσεις μεταβλητών ξεκινούν με τον τύπο. Ακολουθούν ένα ή περισσότερα ονόματα μεταβλητών,
12 χωρισμένα με κόμματα. Παραδείγματα δηλώσεων είναι:

```
13 int i  
14 int x, y, z  
15 char[] s
```

16 1.3.2 Δομικές μονάδες

17 Ο ορισμός μίας δομικής μονάδας γίνεται με τη λέξη κλειδί `def`, που ακολουθείται από την επικεφαλίδα
18 της δομικής μονάδας, τις τοπικές δηλώσεις και το σώμα της. Στην επικεφαλίδα αναφέρεται ο τύπος επι-
19 στροφής (προαιρετικά), το όνομα της δομικής μονάδας και οι τυπικές της παράμετροι (προαιρετικά). Ο
20 τύπος επιστροφής παραλείπεται για τις δομικές μονάδες που δεν επιστρέφουν αποτέλεσμα (πρβλ. διαδι-
21 κασίες στην Pascal). Οι τυπικές παράμετροι γράφονται μέσα σε παρενθέσεις.

22 Κάθε τυπική παράμετρος χαρακτηρίζεται από το όνομά της, τον τύπο της και τον τρόπο περάσματος.
23 Η δήλωση των παραμέτρων μοιάζει με εκείνη των μεταβλητών. Συνεχόμενες δηλώσεις παραμέτρων με
24 διαφορετικούς τύπους ή τρόπο περάσματος διαχωρίζονται μεταξύ τους με ελληνικά ερωτηματικά (;). Η
25 γλώσσα Tony υποστηρίζει πέρασμα παραμέτρων κατ' αξία (by value) και κατ' αναφορά (by reference).
26 Αν η δήλωση ξεκινά με τη λέξη κλειδί `ref`, τότε οι παράμετροι που δηλώνονται περνούν κατ' αναφορά,
27 διαφορετικά περνούν κατ' αξία.

28 Ακολουθούν παραδείγματα επικεφαλίδων ορισμών δομικών μονάδων.

```
29 def p1 ()  
30 def p2 (int n)  
31 def p3 (int a, b; ref char c)  
32 def int f1 (int x)  
33 def int f2 (char[] s)  
34 def int[][] matrix_mult(int p, q, r; int[][] a, b)
```

35 Οι τοπικές δηλώσεις μιας δομικής μονάδας ακολουθούν την επικεφαλίδα. Η Tony ακολουθεί τους κα-
36 νόνες εμφάνισης της Pascal, όσον αφορά στην ορατότητα των ονομάτων μεταβλητών, δομικών μονάδων
37 και παραμέτρων.

38 Στην περίπτωση αμοιβαία αναδρομικών υποπρογραμμάτων, το όνομα ενός υποπρογράμματος χρειά-
39 ζεται να εμφανιστεί πριν τον ορισμό της. Στην περίπτωση αυτή, για να μην παραβιαστούν οι κανόνες
40 εμφάνισης, πρέπει να έχει προηγηθεί μια δήλωση της επικεφαλίδας αυτού του υποπρογράμματος, χωρίς
41 το σώμα του. Αυτό γίνεται με τη λέξη κλειδί `decl` αντί της `def`.

1 1.4 Εκφράσεις

2 Κάθε έκφραση της Tony διαθέτει ένα μοναδικό τύπο και μπορεί να αποτιμηθεί δίνοντας ως αποτέλεσμα
3 μια τιμή αυτού του τύπου. (Μοναδική εξαίρεση είναι η σταθερά nil, η οποία έχει τύπο λίστας οποιου-
4 δήποτε τύπου.) Οι εκφράσεις διακρίνονται σε δύο κατηγορίες: αυτές που δίνουν l-values, οι οποίες περι-
5 γράφονται στην ενότητα 1.4.1 και αυτές που δίνουν r-values, που περιγράφονται στις ενότητες 1.4.2 ως
6 1.4.4. Τα δυο αυτά είδη τιμών έχουν πάρει το όνομά τους από τη θέση τους σε μια εντολή ανάθεσης: οι
7 l-values εμφανίζονται στο αριστερό μέλος της ανάθεσης ενώ οι r-values στο δεξιό.

8 Οι εκφράσεις μπορούν να εμφανίζονται μέσα σε παρενθέσεις, που χρησιμοποιούνται για λόγους ομα-
9 δοποίησης.

10 1.4.1 L-values

11 Οι l-values αντιπροσωπεύουν αντικείμενα που καταλαμβάνουν χώρο στη μνήμη του υπολογιστή κατά
12 την εκτέλεση του προγράμματος και τα οποία μπορούν να περιέχουν τιμές. Τέτοια αντικείμενα είναι οι
13 μεταβλητές και οι παράμετροι των δομικών μονάδων και τα στοιχεία πινάκων. Συγκεκριμένα:

- 14 • Το όνομα μιας μεταβλητής ή μιας παραμέτρου συνάρτησης είναι l-value και αντιστοιχεί στο εν
15 λόγω αντικείμενο. Ο τύπος της l-value είναι ο τύπος του αντίστοιχου αντικειμένου.
- 16 • Αν e_1 είναι μια έκφραση τύπου t [] και e_2 είναι μια έκφραση τύπου int , τότε $e_1[e_2]$ είναι μια
17 l-value με τύπο t . Αν η τιμή της έκφρασης e_2 είναι ο μη αρνητικός ακέραιος n τότε αυτή η l-
18 value αντιστοιχεί στο στοιχείο με δείκτη n του πίνακα που αντιστοιχεί στην e_1 . Η αρίθμηση των
19 στοιχείων του πίνακα ξεκινά από το μηδέν. Η τιμή του n δεν πρέπει να υπερβαίνει τα πραγματικά
20 όρια του πίνακα.

21 Αν μια l-value χρησιμοποιηθεί ως έκφραση, η τιμή της είναι ίση με την τιμή που περιέχεται στο
22 αντικείμενο που αντιστοιχεί σε αυτήν.

23 1.4.2 Σταθερές

24 Στις r-values της γλώσσας Tony συγκαταλέγονται οι ακόλουθες σταθερές:

- 25 • Οι ακέραιες σταθερές χωρίς πρόσημο, όπως περιγράφονται στην ενότητα 1.1. Έχουν τύπο int και
26 η τιμή τους είναι ίση με τον μη αρνητικό ακέραιο αριθμό που παριστάνουν.
- 27 • Οι σταθεροί χαρακτήρες, όπως περιγράφονται στην ενότητα 1.1. Έχουν τύπο char και η τιμή τους
28 είναι ίση με το χαρακτήρα που παριστάνουν.
- 29 • Οι σταθερές συμβολοσειρές, όπως περιγράφονται στην ενότητα 1.1. Έχουν τύπο $\text{char}[]$. Κάθε
30 τέτοια έκφραση αντιστοιχεί σε έναν πίνακα, στον οποίο βρίσκονται αποθηκευμένοι οι χαρακτήρες
31 της συμβολοσειράς. Στο τέλος του πίνακα αποθηκεύεται αυτόματα ο χαρακτήρας $'\backslash 0'$, σύμφωνα
32 με τη σύμβαση που ακολουθεί η γλώσσα C για τις συμβολοσειρές. Οι σταθερές συμβολοσειρές
33 είναι το μόνο είδος σταθεράς τύπου πίνακα που επιτρέπεται.
- 34 • Οι λέξεις κλειδιά `true` και `false`, που έχουν τύπο bool .
- 35 • Η λέξη κλειδί `nil`, που παριστάνει την κενή λίστα και έχει τύπο $\text{list}[t]$ για κάθε έγκυρο τύπο t .

36 1.4.3 Τελεστές

37 Οι τελεστές της Tony διακρίνονται σε τελεστές με ένα ή δύο τελούμενα. Οι τελεστές με ένα τελούμενο
38 γράφονται πριν από αυτό (prefix), ενώ οι τελεστές με δύο τελούμενα γράφονται πάντα μεταξύ των τε-
39 λουμένων (infix). Η αποτίμηση των τελουμένων γίνεται από αριστερά προς τα δεξιά. Οι τελεστές με δύο
40 τελούμενα αποτιμούν υποχρεωτικά και τα δύο τελούμενα, με εξαίρεση τους τελεστές `and` και `or`, όπως
41 περιγράφεται παρακάτω. Υπάρχουν επίσης τέσσερις τελεστές με ειδική σύνταξη. Όλοι οι τελεστές της
42 Tony έχουν ως αποτέλεσμα r-value.

- Οι τελεστές με ένα τελούμενο `+` και `-` υλοποιούν τους τελεστές προσήμου. Το τελούμενο πρέπει να είναι έκφραση τύπου `int` και το αποτέλεσμα είναι r-value του ίδιου τύπου.
- Ο τελεστής με ένα τελούμενο `not` υλοποιεί τη λογική άρνηση. Το τελούμενο πρέπει να είναι έκφραση τύπου `bool` και το ίδιο είναι το αποτέλεσμά του.
- Οι τελεστές με δύο τελούμενα `+`, `-`, `*`, `/` και `mod` υλοποιούν τις αριθμητικές πράξεις. Τα τελούμενα πρέπει να είναι εκφράσεις τύπου `int` και το αποτέλεσμα είναι r-value του ίδιου τύπου.
- Οι τελεστές `=`, `<>`, `<`, `>`, `<=` και `>=` υλοποιούν τις σχέσεις σύγκρισης μεταξύ βασικών τύπων. Τα τελούμενα πρέπει να είναι εκφράσεις του ίδιου βασικού τύπου t και το αποτέλεσμα είναι τύπου `bool`.
- Οι τελεστές `and` και `or` υλοποιούν αντίστοιχα τις πράξεις της λογικής σύζευξης και διάζευξης. Τα τελούμενα πρέπει να είναι εκφράσεις τύπου `bool` και το ίδιο είναι και το αποτέλεσμα. Η αποτίμηση συνθηκών που χρησιμοποιούν αυτούς τους τελεστές γίνεται με *βραχυκύκλωση* (short-circuit). Δηλαδή, αν το αποτέλεσμα της συνθήκης είναι γνωστό από την αποτίμηση και μόνο του πρώτου τελούμενου, το δεύτερο τελούμενο δεν αποτιμάται καθόλου.
- Ο τελεστής `#` υλοποιεί την κατασκευή μη κενής λίστας (cons). Το πρώτο τελούμενο πρέπει να είναι έκφραση κάποιου έγκυρου τύπου t — η κεφαλή — και το δεύτερο τελούμενο πρέπει να είναι έκφραση τύπου `list[t]` — η ουρά. Το αποτέλεσμα είναι επίσης τύπου `list[t]`.
- Οι τελεστές με ένα τελούμενο `nil?`, `head` και `tail` υλοποιούν τρεις βασικές πράξεις λιστών. Η σύνταξή τους μοιάζει με κλήση συναρτήσεων. Το τελούμενο γράφεται μέσα σε παρενθέσεις και πρέπει να είναι τύπου `list[t]`, για κάποιον έγκυρο τύπο t . Ο τελεστής `nil?` επιστρέφει αποτέλεσμα τύπου `bool`: ελέγχει αν μια λίστα είναι κενή ή όχι. Ο τελεστής `head` επιστρέφει αποτέλεσμα τύπου t : την κεφαλή της λίστας. Ο τελεστής `tail` επιστρέφει αποτέλεσμα τύπου `list[t]`: την ουρά της λίστας. Οι τελεστές `head` και `tail` δεν πρέπει να καλούνται με λίστες οι οποίες είναι κενές.
- Ο τελεστής `new` χρησιμοποιείται για τη δημιουργία πινάκων και έχει την ειδική σύνταξη `new t [e]`. Το t πρέπει να είναι ένας έγκυρος τύπος και η έκφραση e πρέπει να έχει τύπο `int`. Η τιμή της e πρέπει να είναι κάποιος θετικός ακέραιος n . Το αποτέλεσμα έχει τύπο `t []` και είναι ένας νέος πίνακας n στοιχείων, των οποίων οι αρχικές τιμές είναι απροσδιόριστες.

Στον πίνακα 2 ορίζεται η προτεραιότητα και η προσεταιριστικότητα των τελεστών της Tony. Οι γραμμές που βρίσκονται υψηλότερα στον πίνακα περιέχουν τελεστές μεγαλύτερης προτεραιότητας. Τελεστές που βρίσκονται στην ίδια γραμμή έχουν την ίδια προτεραιότητα. Οι τελεστές `nil?`, `head`, `tail` και `new`, που έχουν ειδική σύνταξη, δεν συμπεριλαμβάνονται στον πίνακα.

1.4.4 Κλήση δομικών μονάδων ως συναρτήσεων

Αν f είναι το όνομα μιας δομικής μονάδας με υπαρκτό τύπο επιστροφής t , τότε η έκφραση $f(e_1, \dots, e_n)$ είναι μια r-value με τύπο t . Ο αριθμός των πραγματικών παραμέτρων n πρέπει να συμπίπτει με τον αριθμό των τυπικών παραμέτρων της f . Επίσης, ο τύπος και το είδος κάθε πραγματικής παραμέτρου πρέπει να συμπίπτει με τον τύπο και τον τρόπο περάσματος της αντίστοιχης τυπικής παραμέτρου, σύμφωνα με τους παρακάτω κανόνες.

- Αν η τυπική παράμετρος είναι τύπου t και περνά κατ' αξία, τότε η αντίστοιχη πραγματική παράμετρος πρέπει να είναι έκφραση τύπου t .
- Αν η τυπική παράμετρος είναι τύπου t και περνά κατ' αναφορά, τότε η αντίστοιχη πραγματική παράμετρος πρέπει να είναι l-value τύπου t .

Κατά την κλήση μιας δομικής μονάδας, οι πραγματικές παράμετροι αποτιμώνται από αριστερά προς τα δεξιά.

Πίνακας 2: Προτεραιότητα και προσεταιριστικότητα των τελεστών της Tony.

Τελεστές	Περιγραφή	Αριθμός τελουμένων	Θέση και προσεταιριστικότητα
+ -	Πρόσημα	1	prefix
* / mod	Πολλαπλασιαστικοί τελεστές	2	infix, αριστερή
+ -	Προσθετικοί τελεστές	2	infix, αριστερή
#	Κατασκευή λίστας	2	infix, δεξιά
= <> > < <= >=	Σχεσιακοί τελεστές	2	infix, καμία
not	Λογική άρνηση	1	prefix
and	Λογική σύζευξη	2	infix, αριστερή
or	Λογική διάζευξη	2	infix, αριστερή

1.5 Εντολές

Οι εντολές που υποστηρίζει η γλώσσα Tony χωρίζονται σε απλές και σύνθετες. Οι απλές εντολές είναι οι ακόλουθες:

- Η κενή εντολή `skip` που δεν κάνει καμία ενέργεια.
- Η εντολή ανάθεσης $\ell := e$ που αναθέτει την τιμή της έκφρασης e στην l-value ℓ . Η l-value ℓ και η έκφραση e πρέπει να είναι του ίδιου τύπου t . Η l-value ℓ δεν πρέπει να αντιστοιχεί σε στοιχείο σταθερής συμβολοσειράς. Σύμφωνα με όσα γράφονται στην ενότητα 1.2, η ανάθεση πινάκων και λιστών αντιγράφει τη διεύθυνση ενός πίνακα ή μίας λίστας, δημιουργώντας συνώνυμα — ποτέ δεν αντιγράφει τα περιεχόμενα.
- Η εντολή κλήσης δομικής μονάδας. Οι πραγματικές παράμετροι γράφονται ανάμεσα σε παρενθέσεις. Αν είναι περισσότερες, χωρίζονται με κόμματα. Οι προϋποθέσεις για την κλήση είναι οι ίδιες όπως στην ενότητα 1.4.4 με τη διαφορά ότι δεν πρέπει να ορίζεται τύπος επιστροφής για τη δομική μονάδα.

Οι σύνθετες εντολές της Tony είναι οι εξής:

- Η εντολή ελέγχου `if e_1 : b_1 elsif e_2 : b_2 else : b_3` , όπου το σκέλος `elsif` μπορεί να επαναλαμβάνεται μηδέν ή περισσότερες φορές και το σκέλος `else` είναι προαιρετικό. Τα e_1 και e_2 πρέπει να είναι έγκυρες εκφράσεις τύπου `bool`. Τα b_1 , b_2 και b_3 πρέπει να είναι ακολουθίες μίας ή περισσότερων εντολών. Η σημασιολογία της εντολής είναι όπως π.χ. στη Python.
- Η εντολή ελέγχου `for s_1 ; e ; s_2 : b` , που μοιάζει με την αντίστοιχη εντολή της C. Τα s_1 και s_2 είναι ακολουθίες μίας ή περισσότερων απλών εντολών, χωρισμένων μεταξύ τους με κόμματα. Το s_1 εκτελείται μία φορά στην έναρξη του βρόχου (αρχικοποίηση), ενώ το s_2 εκτελείται στο τέλος κάθε επανάληψης (βήμα). Η έκφραση e πρέπει να είναι τύπου `bool` και παριστάνει τη συνθήκη του βρόχου: όσο η τιμή της είναι αληθής, ο βρόχος επαναλαμβάνεται. Το b πρέπει να είναι ακολουθία μίας ή περισσότερων εντολών.
- Η εντολή άλματος `exit` που τερματίζει την εκτέλεση της τρέχουσας δομικής μονάδας. Η εντολή αυτή πρέπει να εμφανίζεται στο σώμα μίας δομικής μονάδας χωρίς τύπο επιστροφής.
- Η εντολή άλματος `return e` που τερματίζει την εκτέλεση της τρέχουσας δομικής μονάδας και επιστρέφει ως αποτέλεσμα την τιμή της έκφρασης e . Η εντολή αυτή πρέπει να εμφανίζεται στο σώμα μίας δομικής μονάδας με τύπο επιστροφής t και η έκφραση e πρέπει να έχει τον ίδιο τύπο t .

1 1.6 Βιβλιοθήκη έτοιμων συναρτήσεων

2 Η Tony υποστηρίζει ένα σύνολο προκαθορισμένων δομικών μονάδων, οι οποίες έχουν υλοποιηθεί σε
3 assembly του x86 ως μια βιβλιοθήκη έτοιμων συναρτήσεων (run-time library). Είναι ορατές σε κάθε δο-
4 μική μονάδα, εκτός αν επισκιάζονται από μεταβλητές, παραμέτρους ή άλλες δομικές μονάδες με το ίδιο
5 όνομα. Παρακάτω δίνονται οι δηλώσεις τους και εξηγείται η λειτουργία τους.

6 1.6.1 Είσοδος και έξοδος

```
7 decl puti (int n)  
8 decl putb (bool b)  
9 decl putc (char c)  
10 decl puts (char[] s)
```

11 Οι συναρτήσεις αυτές χρησιμοποιούνται για την εκτύπωση τιμών που ανήκουν στους βασικούς τύπους
12 της Tony, καθώς και για την εκτύπωση συμβολοσειρών.

```
13 decl int geti ()  
14 decl bool getb ()  
15 decl char getc ()  
16 decl gets (int n, char[] s)
```

17 Αντίστοιχα, οι παραπάνω συναρτήσεις χρησιμοποιούνται για την εισαγωγή τιμών που ανήκουν στους
18 βασικούς τύπους της Tony και για την εισαγωγή συμβολοσειρών. Η συνάρτηση gets χρησιμοποιείται για
19 την ανάγνωση μιας συμβολοσειράς μέχρι τον επόμενο χαρακτήρα αλλαγής γραμμής. Οι παράμετροι της
20 καθορίζουν το μέγιστο αριθμό χαρακτήρων (συμπεριλαμβανομένου του τελικού '\0') που επιτρέπεται
21 να διαβαστούν και τον πίνακα χαρακτήρων στον οποίο αυτοί θα τοποθετηθούν. Ο χαρακτήρας αλλαγής
22 γραμμής δεν αποθηκεύεται. Αν το μέγεθος του πίνακα εξαντληθεί πριν συναντηθεί χαρακτήρας αλλαγής
23 γραμμής, η ανάγνωση θα συνεχιστεί αργότερα από το σημείο όπου διακόπηκε.

24 1.6.2 Συναρτήσεις μετατροπής

```
25 decl int abs (int n)  
26 decl int ord (char c)  
27 decl char chr (int n)
```

28 Η συνάρτηση abs υπολογίζει την απόλυτη τιμή ενός ακέραιου αριθμού. Οι συναρτήσεις ord και chr
29 μετατρέπουν από ένα χαρακτήρα στον αντίστοιχο κωδικό ASCII και αντίστροφα.

30 1.6.3 Συναρτήσεις διαχείρισης συμβολοσειρών

```
31 decl int strlen (char[] s)  
32 decl int strcmp (char[] s1, s2)  
33 decl strcpy (char[] trg, src)  
34 decl strcat (char[] trg, src)
```

35 Οι συναρτήσεις αυτές έχουν ακριβώς την ίδια λειτουργία με τις συνώνυμες τους στη βιβλιοθήκη συναρ-
36 τήσεων της γλώσσας C.

2 Πλήρης γραμματική της Tony

Η σύνταξη της γλώσσας Tony δίνεται παρακάτω σε μορφή EBNF. Η γραμματική που ακολουθεί είναι *διφορούμενη*, οι περισσότερες αμφισημίες όμως μπορούν να ξεπεραστούν αν λάβει κανείς υπόψη τους κανόνες προτεραιότητας και προσηταιριστικότητας των τελεστών, όπως περιγράφονται στον πίνακα 2. Τα σύμβολα <id>, <int-const>, <char-const> και <string-literal> είναι τερματικά σύμβολα της γραμματικής.

```

1 <program> ::= <func-def>
2 <func-def> ::= "def" <header> ":" ( ( <func-def> | <func-decl> | <var-def> ) ) * ( <stmt> ) + "end"
3 <header> ::= [ <type> ] <id> "(" [ <formal> ( ( ";" <formal> ) ) * ] ")"
4 <formal> ::= [ "ref" ] <type> <id> ( ( "," <id> ) ) *
5 <type> ::= "int" | "bool" | "char" | <type> "[" "]" | "list" "[" <type> "]"
6 <func-decl> ::= "decl" <header>
7 <var-def> ::= <type> <id> ( ( "," <id> ) ) *
8 <stmt> ::= <simple> | "exit" | "return" <expr>
9 | "if" <expr> ":" ( <stmt> ) + ( ( "elseif" <expr> ":" ( <stmt> ) + ) ) *
10 | [ "else" ":" ( <stmt> ) + ] "end"
11 | "for" <simple-list> ";" <expr> ";" <simple-list> ":" ( <stmt> ) + "end"
12 <simple> ::= "skip" | <atom> ":" <expr> | <call>
13 <simple-list> ::= <simple> ( ( "," <simple> ) ) *
14 <call> ::= <id> "(" [ <expr> ( ( "," <expr> ) ) * ] ")"
15 <atom> ::= <id> | <string-literal> | <atom> "[" <expr> "]" | <call>
16 <expr> ::= <atom> | <int-const> | <char-const> | "(" <expr> ")"
17 | ("+" | "-") <expr> | <expr> ("+" | "-" | "*" | "/" | "mod") <expr>
18 | <expr> ("=" | "<" | ">" | "<=" | ">=") <expr>
19 | "true" | "false" | "not" <expr> | <expr> ("and" | "or") <expr>
20 | "new" <type> "[" <expr> "]" | "nil" | "nil?" "(" <expr> ")"
21 | <expr> "#" <expr> | "head" "(" <expr> ")" | "tail" "(" <expr> ")"
22

```

3 Παραδείγματα

Στην παράγραφο αυτή δίνονται έξι παραδείγματα προγραμμάτων στη γλώσσα Tony, η πολυπλοκότητα των οποίων κυμαίνεται σημαντικά. Για κάποια από αυτά τα παραδείγματα (ή για παρόμοια προγράμματα), μπορείτε να βρείτε τον αρχικό, τον ενδιάμεσο κώδικα (χωρίς βελτιστοποίηση), τη μορφή των εγγραφήματων δραστηριοποίησης των δομικών μονάδων, καθώς και τον τελικό κώδικα σε αντίστοιχα φυλλάδια περιγραφής γλωσσών που δόθηκαν ως θέματα εργασίας στο ίδιο μάθημα σε προηγούμενα έτη, μέσω της ιστοσελίδας του μαθήματος.

3.1 Πες γεια!

Το παρακάτω παράδειγμα είναι το απλούστερο πρόγραμμα στη γλώσσα Tony που παράγει κάποιο αποτέλεσμα ορατό στο χρήστη. Το πρόγραμμα αυτό τυπώνει απλώς ένα μήνυμα.

```


1 def hello ():
2     puts("Hello world!\n")
3 end

```

3.2 Οι πύργοι του Hanoi

Το πρόγραμμα που ακολουθεί λύνει το πρόβλημα των πύργων του Hanoi. Μια σύντομη περιγραφή του προβλήματος δίνεται παρακάτω.

Υπάρχουν τρεις στύλοι, στον πρώτο από τους οποίους είναι περασμένοι n το πλήθος δακτύλιοι. Οι εξωτερικές διαμέτροι των δακτυλίων είναι διαφορετικές και αυτοί είναι περασμένοι από κάτω προς τα πάνω σε φθίνουσα σειρά εξωτερικής διαμέτρου, όπως φαίνεται στο σχήμα 1. Ζητείται να μεταφερθούν



Σχήμα 1: Οι πύργοι του Hanoi.

οι δακτύλιοι από τον πρώτο στον τρίτο στύλο (χρησιμοποιώντας το δεύτερο ως βοηθητικό χώρο), ακολουθώντας όμως τους εξής κανόνες:

- Κάθε φορά επιτρέπεται να μεταφερθεί ένας μόνο δακτύλιος, από κάποιο στύλο σε κάποιον άλλο στύλο.
- Απαγορεύεται να τοποθετηθεί δακτύλιος με μεγαλύτερη διάμετρο πάνω από δακτύλιο με μικρότερη διάμετρο.

Το πρόγραμμα στη γλώσσα Tony που λύνει αυτό το πρόβλημα δίνεται παρακάτω. Η συνάρτηση hanoi είναι αναδρομική.

```

1  def solve ():
2
3      def hanoi (int rings; char[] source, target, auxiliary):
4
5          def move (char[] source, target):
6              puts("Moving from ") puts(source) puts(" to ") puts(target) puts(".\n")
7              end
8
9          if rings >= 1:
10             hanoi(rings-1, source, auxiliary, target)
11             move(source, target)
12             hanoi(rings-1, auxiliary, target, source)
13         end
14     end
15
16     int NumberOfRings
17
18     puts("Rings: ")
19     NumberOfRings := geti()
20     hanoi(NumberOfRings, "left", "right", "middle")
21 end

```

3.3 Πρώτοι αριθμοί

Το παρακάτω παράδειγμα προγράμματος στη γλώσσα Tony είναι ένα πρόγραμμα που υπολογίζει τους πρώτους αριθμούς μεταξύ 1 και n , όπου το n καθορίζεται από το χρήστη. Το πρόγραμμα αυτό χρησιμοποιεί έναν απλό αλγόριθμο για τον υπολογισμό των πρώτων αριθμών. Μια διατύπωση αυτού του αλγορίθμου σε ψευδογλώσσα δίνεται παρακάτω. Λαμβάνεται υπόψη ότι οι αριθμοί 2 και 3 είναι πρώτοι, και στη συνέχεια εξετάζονται μόνο οι αριθμοί της μορφής $6k \pm 1$, όπου k φυσικός αριθμός.

Κύριο πρόγραμμα

τύπωσε τους αριθμούς 2 και 3

για $t := 6$ μέχρι n με βήμα 6 κάνε τα εξής:

αν ο αριθμός $t - 1$ είναι πρώτος τότε τύπωσε τον

αν ο αριθμός $t + 1$ είναι πρώτος τότε τύπωσε τον

Αλγόριθμος ελέγχου (είναι ο αριθμός t πρώτος;)

αν $t < 0$ τότε έλεγξε τον αριθμό $-t$

αν $t < 2$ τότε ο t δεν είναι πρώτος

αν $t = 2$ τότε ο t είναι πρώτος

αν ο t διαιρείται με το 2 τότε ο t δεν είναι πρώτος

για $i := 3$ μέχρι $t/2$ με βήμα 2 κάνε τα εξής:

αν ο t διαιρείται με τον i τότε ο t δεν είναι πρώτος

ο t είναι πρώτος

Το αντίστοιχο πρόγραμμα στη γλώσσα Tony είναι το ακόλουθο.

```
1 def main ():
2
3     def bool prime? (int n):
4         int i
5
6         if n < 0:           return prime?(-n)
7         elsif n < 2:       return false
8         elsif n = 2:       return true
9         elsif n mod 2 = 0: return false
10        else:
11            for i := 3; i <= n / 2; i := i + 2:
12                if n mod i = 0: return false end
13            end
14            return true
15        end
16    end
17
18    int limit, number, counter
19
20    puts("Limit: ")
21    limit := geti()
22    puts("Primes:\n")
23    counter := 0
24    if limit >= 2: counter := counter + 1 puti(2) puts("\n") end
25    if limit >= 3: counter := counter + 1 puti(3) puts("\n") end
26    for number := 6; number <= limit; number := number + 6:
27        if prime?(number - 1):
28            counter := counter + 1
29            puti(number - 1)
30            puts("\n")
31        end
32        if number <> limit and prime?(number + 1):
33            counter := counter + 1
34            puti(number + 1)
35            puts("\n")
36        end
37    end
38
39    puts("\nTotal: ") puti(counter) puts("\n")
40 end
```

3.4 Αντιστροφή συμβολοσειράς

Το πρόγραμμα που ακολουθεί στη γλώσσα Tony εκτυπώνει το μήνυμα “Hello world!” αντιστρέφοντας τη δοθείσα συμβολοσειρά.

```
1 def main ():
2
3     def char[] reverse (char[] s):
4         char[] t
5         int i, l
6
7         l := strlen(s)
8         t := new char[l+1]
9         for i := 0; i < l; i := i+1: t[i] := s[l-i-1] end
10        t[i] := '\0'
11        return t
12    end
13
14    puts(reverse("\n!dlrow olleH"))
15 end
```

3.5 Ταξινόμηση πίνακα με τη μέθοδο της φυσαλίδας

Ο αλγόριθμος της φυσαλίδας (bubble sort) είναι ένας από τους πιο γνωστούς και απλούς αλγορίθμους ταξινόμησης. Το παρακάτω πρόγραμμα σε Tony τον χρησιμοποιεί για να ταξινομήσει έναν πίνακα ακεραίων αριθμών κατ’ αύξουσα σειρά. Αν x είναι ο πίνακας που πρέπει να ταξινομηθεί και n είναι το μέγεθός του (θεωρούμε σύμφωνα με τη σύμβαση της Tony ότι τα στοιχεία του είναι τα $x[0], x[1], \dots, x[n-1]$), μια παραλλαγή του αλγορίθμου περιγράφεται με ψευδοκώδικα ως εξής:

Αλγόριθμος της φυσαλίδας (bubble sort)

επανάλαβε το εξής:

για i από 0 ως $n-2$

αν $x[i] > x[i+1]$

αντίστρεψε τα $x[i]$ και $x[i+1]$

όσο μεταβάλλεται η σειρά των στοιχείων του x

1 Το αντίστοιχο πρόγραμμα σε γλώσσα Tony είναι το εξής:

```
2 def main ():
3
4     def bsort (int n; int[] x):
5
6         def swap (ref int x, y):
7             int t
8             t := x
9             x := y
10            y := t
11            end
12
13            int i
14            bool changed
15
16            for changed := true; changed; skip:
17                changed := false
18                for i := 0; i < n-1; i := i+1:
19                    if x[i] > x[i+1]:
20                        swap(x[i], x[i+1])
```

```

21         changed := true
22     end
23 end
24 end
25 end
26
27 def writeArray (char[] msg; int n; int[] x):
28     int i
29
30     puts(msg)
31     for i := 0; i < n; i := i+1:
32         if i > 0: puts(", ") end
33         puti(x[i])
34     end
35     puts("\n")
36 end
37
38 int seed, i
39 int[] x
40
41 x := new int[16]
42 seed := 65
43 for i := 0; i < 16; i := i+1:
44     seed := (seed * 137 + 220 + i) mod 101
45     x[i] := seed
46 end
47 writeArray("Initial array: ", 16, x)
48 bsort(16, x)
49 writeArray("Sorted array: ", 16, x)
50 end

```

3.6 Ταξινόμηση λίστας με τη μέθοδο της διαμέρισης

Ο αλγόριθμος της διαμέρισης (quicksort) δε θα μπορούσε φυσικά να λείπει από το εγχειρίδιο ορισμού της γλώσσας Tony! Η (σχεδόν) συναρτησιακή παραλλαγή του για την ταξινόμηση λιστών, που περιγράφεται παρακάτω, είναι διαφορετική από την προστακτική πρωτότυπη μορφή που οφείλεται στον Tony Hoare και που ταξινομεί έναν πίνακα επί τόπου.

Αλγόριθμος της διαμέρισης (quicksort)

χρησιμοποίησε το πρώτο στοιχείο της λίστας ως διαμεριστή (pivot)
 χώρισε τα υπόλοιπα στοιχεία της λίστας σε δύο λίστες:
 μία που περιέχει τα στοιχεία που είναι μικρότερα του διαμεριστή, και
 μία που περιέχει τα στοιχεία που είναι μεγαλύτερα ή ίσα
 ταξινόμησε ξεχωριστά (αναδρομικά) τις δύο λίστες
 συνένωσε τα αποτελέσματα, παρεμβάλλοντας το διαμεριστή

1 Αντί της ξεχωριστής ταξινόμησης και συνένωσης, χρησιμοποιείται η συνάρτηση `qsort_aux(l, r)` που
 2 ταξινομεί τη λίστα `l` και συγχρόνως συνενώνει στο αποτέλεσμα τη λίστα `r`. Το πρόγραμμα σε γλώσσα
 3 Tony είναι το εξής:

```

4 def main ():
5
6     def list[int] qsort (list[int] l):
7
8         def list[int] qsort_aux (list[int] l, rest):
9             int pivot, x
10            list[int] lt, ge

```



```

11
12     if nil?(l): return rest end
13     pivot := head(l)
14     l := tail(l)
15     for lt := nil, ge := nil; not nil?(l); l := tail(l):
16         x := head(l)
17         if x < pivot: lt := x # lt else: ge := x # ge end
18     end
19     return qsort_aux(lt, pivot # qsort_aux(ge, rest))
20 end
21
22     return qsort_aux(l, nil)
23 end
24
25 def writeList (char[] msg; list[int] l):
26     bool more
27
28     puts(msg)
29     for more := false; not nil?(l); l := tail(l), more := true:
30         if more: puts(", ") end
31         puti(head(l))
32     end
33     puts("\n")
34 end
35
36 int seed, i
37 list[int] l
38
39 seed := 65
40 for i := 0, l := nil; i < 16; i := i+1:
41     seed := (seed * 137 + 220 + i) mod 101
42     l := seed # l
43 end
44 writeList("Initial list: ", l)
45 l := qsort(l)
46 writeList("Sorted list: ", l)
47 end

```

4 Οδηγίες για την παράδοση

Ο τελικός μεταγλωττιστής πρέπει να μπορεί να εξάγει κατά βούληση ενδιάμεσο και τελικό κώδικα. Εφόσον δεν έχουν καθορισθεί οι παράμετροι λειτουργίας `-f` ή `-i`, που εξηγούνται παρακάτω, ο μεταγλωττιστής θα δέχεται το πηγαίο πρόγραμμα από ένα αρχείο με οποιαδήποτε κατάληξη (π.χ. `*.tony`) που θα δίνεται ως το μοναδικό του όρισμα. Ο ενδιάμεσος κώδικας θα τοποθετείται σε αρχείο με κατάληξη `*.imm` και ο τελικός σε αρχείο με κατάληξη `*.asm`. Τα αρχεία αυτά θα βρίσκονται στον ίδιο κατάλογο και θα έχουν το ίδιο κυρίως όνομα. Π.χ. από το πηγαίο αρχείο `/tmp/hello.tony` θα παράγονται τα `/tmp/hello.imm` και `/tmp/hello.asm`.

Το εκτελέσιμο του τελικού μεταγλωττιστή θα πρέπει να δέχεται τις παρακάτω παραμέτρους:

- O σημαία βελτιστοποίησης (προαιρετική).
- f πρόγραμμα στο standard input, έξοδος τελικού κώδικα στο standard output.
- i πρόγραμμα στο standard input, έξοδος ενδιάμεσου κώδικα στο standard output.

Επίσης, η τιμή που θα επιστρέφεται στο λειτουργικό σύστημα από το μεταγλωττιστή θα πρέπει να είναι μηδενική στην περίπτωση επιτυχούς μεταγλώττισης και μη μηδενική σε αντίθετη περίπτωση.

Για την εύκολη ανάπτυξη του μεταγλωττιστή προτείνεται η χρήση ενός αρχείου `Makefile` με τα εξής (τουλάχιστον) χαρακτηριστικά:

- Με απλό `make`, θα δημιουργεί το εκτελέσιμο του τελικού μεταγλωττιστή.
- Με `make clean`, θα σβήνει όλα ανεξαιρέτως τα αρχεία που παράγονται αυτόματα (π.χ. αυτά που παράγουν `bison` και `flex`, τα object files και το τελικό εκτελέσιμο).
- Με `make distclean`, θα σβήνει όλα τα παραπάνω και το τελικό εκτελέσιμο.

Ένα παράδειγμα ενός τέτοιου `Makefile`, υποθέτοντας ότι γλώσσα υλοποίησης είναι η C++ και γίνεται χρήση των εργαλείων `flex` και `bison`, είναι το παρακάτω.

```
.PHONY: default clean distclean
CXX=c++
CXXFLAGS=-Wall -std=c++11

default: compiler

lexer.cpp: lexer.l
    flex -s -o lexer.cpp lexer.l

parser.hpp parser.cpp: parser.y
    bison -dv -o parser.cpp parser.y

# Add your dependencies here, e.g.
# parser.o: parser.cpp lexer.hpp symbol.hpp

compiler: lexer.o parser.o symbol.o
    $(CXX) $(CXXFLAGS) -o $@ $^

clean:
    $(RM) lexer.cpp parser.cpp parser.hpp parser.output *.o

distclean: clean
    $(RM) compiler
```